

A Math Game Only a Mother Could Love

© 2012 Gary S. Stager, Ph.D.

Version 1

Create two textboxes on the MicroWorlds page. One should be named, Correct and the other should be named Incorrect.

```
to number1
output random 11
end
```

Number1 will report a random number between 0 and 10. If you want a number from 1-10, you say `output 1 + random 10`

```
to number2
output random 11
end
```

number1 or number2 may be different in case you want to practice one times table or another.

If you wish to practice a particular "table" change the number1 procedure to `output 5`, if you want to practice your 5 times tables.

Try this line a few times and see what it does.
`Show (list number1 "*" number2)`

It should make a multiplication problem

```
to quiz
askquestion (list number1 "*" number2)

end
```

```
to askquestion :problem
question :problem
ifelse answer = run :problem [setcorrect correct + 1]
[setincorrect incorrect + 1]
end
```

Can you add an announcement with ANNOUNCE, sound-effect or animation when the user answers correctly or incorrectly?

```
to game
  setup
  repeat 10 [quiz]
end
```

```
to setup
  setcorrect 0
  setincorrect 0
end
```

Can you figure out a way to randomly select the arithmetic operation [+ - * /] ?
Hint: `PICK` may be useful here.

Can you figure out a way to display a score (perhaps based on percentage of correct answers) on the page? Hint: You'll need a `score` textbox.

`GAME` is the superprocedure that makes everything work. You may wish to make a button to run the `GAME` instruction.

Version 2 - Timed game

change the following procedures

```
to game
  setup
  resett
  repeat 10 [quiz]
end
```

`resett` resets the program's clock to 0.

```
to quiz
  if timer > 600 [Announce [Time's up!] stopall]
  askquestion (list number1 "*" number2)
  quiz
end
```

`timer` counts in tenths of a second. So, 10 = 1 second. 600 = 1 minute. You may use any number you wish in the quiz procedure.

The `quiz` procedure now runs over and over again until the time is up and then `stopall` stops all processes.